><(>< TRICKS OF THE TRADE ><(><
><(><(>< TUTORIAL #15 ><(><(><
><(><(>< MIKE WHITE ><(><(><
><(><(>< COUNTY LINE #2 ><(><(><
><(><(>< R.D.#1 BOX 373 ><(><(><
><(><(>< WAKEMAN , OHIO ><(><(><
><(><(>< 44889 ><(><(><

STILL MORE ON STRINGS

LAST MONTH I SAID THAT WE WOULD BE LOOKING AT A METHOD WHERE DIFFERENT LENGTH SEGMENTS ARE HANDLED IN ONE ARRAY. IT'S CALLED:

THE DIRECTORY METHOD

THIS IS WHAT I FOUND INSIDE **THE GREAT AMERICAN JIGSAW** FROM ESOTERICA LIMITED, P.O. BOX 614, WARREN, OHIO 44482 (IF YOU'VE NEVER SEEN THIS BEFORE, YOU SHOULD! IF YOU GET NO RESULTS BY WRITING TO DAN DREACHER, WRITE ME). THE 1800 BYTE PROGRAM DRAWS ALL 48 CONTINENTAL UNITED STATES ON THE SCREEN. THE PROGRAM IS ABOUT 900 BYTES LONG AND USES %(-X) TO HOLD THE DRAWING STRING. HERE'S THE LINES THAT DO IT!!!

```
490 F=1;L=1
500 E=%(-23676+(Sx2))÷1000;B=RM
505 K=0;J=0;FOR A=-23582+BTO -23582+B+ESTEP 2
510 C=%(A)÷10000;X=RM÷100;Y=RM;C=ABS(C)
520 IF C=1Y=-Y
530 IF F=2IF A=-23582+B K=-52-X;J=-35-Y;L=2
533 X=X+K;Y=Y+J
534 IF A=-23582+BLINE X,Y,4;G=X;H=Y;NEXT A
540 LINE X,Y,L;NEXT A;LINE G,H,L;RETURN
```

THE "DIRECTORY" STARTS AT %(-23680) AND RUNS 96 (48x2) BYTES LONG. THE "FOR NEXT" LOOP IN LINE 505 SETS UP WITH THE DATA FROM THE "DIRECTORY" AND DRAWS ALL 48 STATES USING THE LINE COMMANDS SHOWN. THIS WAY **TEXAS** (25 LINES) AND **RHODE ISLAND** (4 LINES), CAN BE IN THE SAME STRING. THE PART WITH J,K,+L IS SO THE STATE CAN BE DRAWN IN THE LOWER LEFT CORNER OF THE SCREEN BEFORE IT GETS PUT IN THE MAP. THIS IS THE GAME CONCEPT OF THE PROGRAM. PLAYERS 1+2 TRY TO RECOGNIZE STATES BY THIER SHAPE AND LOCATION. "GOSUB 500" PUTS A STATE IN LOWER LEFT, "GOSUB 490" PUTS IT IN THE MAP. F=2 AND S=RND (48) WHEN THIS ROUTINE GETS RUN.

THE GREAT AMERICAN JIGSAW IS THE ONLY BASIC PROGRAM I'VE EVER SEEN USING THIS STRING METHOD. FOR MORE INFORMATION CONSULT **PROGRAMMING THE Z80** BY [RODNAY ZAKS] (TO BE FOUND IN MOST **RADIO SHACK** STORES). THIS METHOD IS COMMON IN MACHINE CODE PROGRAMS.

THE LAST METHOD IS ACTUALLY A HARDWARE METHOD THAT NOBODY (EXCEPT ME) EVER USED. ONLY IN THE 16K VERSION OF **QUADRA** CAN YOU FIND IT, AND IT'S LIMITED TO @(X) ALLOCATIONS ONLY! I CALL IT:

THE PHANTOM-START METHOD

TO DEMONSTRATE, PUT AB IN THE SLOT AND RUN THIS:

```
10 .CD00N GYR0AUT UKLNA0TWI 0WNHSA!T  IYT0 UW 0DRIKDS?!
20 CLEAR ;A=20000;C=%(A);%(A)=-24571;GOSUB 40
30 %(A)=-24570;GOSUB 40;%(A)=C;IF KPRUN
40 FOR B=0TO 25;TV=@(B);NEXT B;RETURN
```

BE SURE TO KEY IN LINE 10 EXACTLY AS SHOWN SPACES AND ALL!!! %(20000) IS THE "TXTUNF" IN AB (SEE PG. 103 AB HANDBOOK). WHEN THIS NUMBER CHANGES (AS WHEN WE CHANGE OUR BASIC TEXT) @(0) GETS MOVED TO A NEW ADDRESS. BY GIVING THE "TXTUNF" A PHANTOM ADDRESS WE START A PHANTOM STRING, EITHER INSIDE THE TEXT (SHOWN ABOVE), OR IN THE FREE SPACE BEYOND. THE PROGRAM CAN BE LISTED, RUN, OR TAPED, BUT NOT EDITED OR CHANGED, UNLESS THE CORRECT "TXTUNF" HAS BEEN RESTORED. "C=%(A)" AND "%(A)=C" SAVES IT IN THE PROGRAM ABOVE. THIS IDEA IS BETTER SEEN IN EB WHERE THE TEXT CAN BE IN %(!6000) ADDRESSES, WITH THE @(X) IN %(!7000) ADDRESSES, OR VICE VERSA. TO RUN THE ABOVE PROGRAM IN BB CHANGE "20000" TO "20050". FOR BRB, CHANGE "20000" TO "27778", AND "-24571" AND "-24570" TO "24579" AND "24580". IN VIPER-SOFT, CHANGE "20000" TO "-32638", AND "-24571" AND "-24570" BECOMES "-31743" AND "-31742". IN LINE 40, "TV=@(B)" PRINTS ONLY THE LOWER BYTE OF THE @(X) LOCATIONS.

ONE LAST NOTE ON "HARDWARE" STRING ALLOCATIONS. CERTAIN DATA IS BEST KEPT IN CERTAIN ALLOCATIONS. SINCE @(X) IS SHORTER THAN "%(X+20258)" IN THE BASIC TEXT, @(X) OR ¥(X) IS BEST FOR KEEPING PLAYERS SCORES, ETCETERA. SINCE ¥(X) DOESN'T CHANGE WITH A LAST MINUTE TEXT CHANGE, IT WORKS BEST FOR "FIXED" STRINGS, AND @(X) WORKS BEST AS AN "AREA". ALSO, MACHINE CODE WON'T RUN FROM @(X), ¥(X), OR %(-X), IN AB AND BB BECAUSE THESE ALLOCATIONS ADDRESSES EXIST ONLY IN THE CARTRIDGE AS NEGATIVE NUMBERS. MACHINE CODE RUNS "AROUND" THE CARTRIDGE, LEAVING US WITH ONLY %(X) OR %(+X), AND %(X) WORKS BEST. HOWEVER, MACHINE CODE CAN RUN FROM ANYWHERE INSIDE EXISTING MEMORY IN EB. (SEE ARCADIAN VOL. 5 PAGES 60, 61, 113, 117, 145, AND 175 FOR MORE INFORMATION).

IN CLOSING THIS SERIES I CAN ONLY REPEAT A LINE I USED IN EVERY ISSUE LAST YEAR. **"IF ANY OF MY PROGRAMS PUZZLE YOU, DON'T HESITATE TO WRITE. A S.A.S.E. SPEEDS UP REPLYS".**

FINALLY, HERE IS A SHORT PROGRAM THAT SHOULD GIVE YOU A GOOD LAUGH!! (**EB ONLY**! SORRY ABOUT THAT):

SILLY-FACE BY MIKE WHITE

```
10 CLEAR ;FOR A=27TO 30;CIRCLE 0,12,A,6;NEXT A;BOX 0,32,160,40,4
20 FOR B=-15TO 15STEP 30;FOR A=7TO 9;CIRCLE B,20,A,5;NEXT A
30 BOX B,20,3,3,7;NEXT B;DATA CX,-33,-32,6,0,L.
40 PRINT "HAPPINESS IS";FOR B=1TO 3000;NEXT B;FOR A=37 TO 40
50 CIRCLE 0,12,A,5;NEXT A;BOX 0,10,9,9,7;DATA CX,-57,-40,7
60 PRINT "A BLUE RAM SYSTEM!!!",;IF KPRUN
```